

ACL2: Recursión e inducción

Francisco J. Martín Mateos
José L. Ruiz Reina

Dpto. Ciencias de la Computación e Inteligencia Artificial
Universidad de Sevilla

Principio de definición

- La lógica se extiende dinámicamente, añadiendo nuevas definiciones de funciones mediante *el principio de definición*, usando `defun`
 - Cada definición añade nuevos axiomas a la lógica
 - La consistencia de la lógica se debe mantener
- Sólo se admiten funciones bajo ciertas restricciones
 - Sólo se puede hacer referencia a funciones previamente definidas
 - Las únicas variables de la definición han de ser las que aparecen como sus argumentos o las introducidas mediante entornos locales
 - Sólo se admiten funciones que terminan para cualquier dato de entrada
- Esto asegura extensiones consistentes y conservativas de la lógica

Principio de definición

- Dada una historia $\mathcal{H}$ (secuencia de definiciones previas), la definición `(defun  $f$  ( $x_1 \dots x_n$ ) cuerpo)` es admisible respecto de $\mathcal{H}$ si
 - f es un símbolo de función nuevo (no aparece en $\mathcal{H}$),
 - cada x_i es un símbolo de variable distinto,
 - `cuerpo` es un término en el lenguaje de $\mathcal{H}$ ampliado con el símbolo f de aridad n , cuyas variables libres están entre las x_i ,
 - se pueden demostrar las *conjeturas de terminación* de f
- Si la definición es aceptada, se introduce el axioma
$$(f\ x_1 \dots x_n) = \text{cuerpo}$$

Principio de definición

- No se permiten definiciones por recursión doble

```
(defun par (n)
  (if (zp n)
      t
      (impar (- n 1))))

(defun impar (n)
  (if (zp n)
      nil
      (par (- n 1))))
```

- La primera definición incluiría un axioma que hace referencia a un símbolo de función desconocido !!

Principio de definición

- No se permite el uso de variables globales

```
(defun g (x)
  y)
```

- Esta definición añadiría el siguiente axioma a la lógica

$$(g\ x) = y$$

donde y es una variable global, cuyo valor puede cambiar después de la definición de g . Este cambio de valor puede hacer que el axioma anterior produzca una inconsistencia en el sistema

Principio de definición

- El propio lenguaje de programación fuerza a que las funciones tomen un valor para cualquier dato de entrada (funciones totales)
 - Todas las funciones predefinidas en el sistema son totales
 - Sólo se permite la versión completa del condicional `if`

```
(if <condicion>  
  <instruccion>  
  <alternativa>)
```

- El caso final del condicional amplio `cond` ha de ser el caso por defecto `t`

```
(cond (<condicion-1> <cuerpo-1>  
      ...  
      (<condicion-n> <cuerpo-n>  
        (t <alternativa>)))
```

- No se permiten las recursiones infinitas

```
(defun f (x)
  (+ 1 (f x)))
```

- Esta definición añadiría el siguiente axioma a la lógica
$$(f\ x) = (+\ 1\ (f\ x))$$
- El uso de esta igualdad como regla de simplificación introduciría el sistema en un bucle infinito

Conjeturas de terminación

- Dada una historia $\mathcal{H}$ (secuencia de definiciones previas), las conjeturas de terminación de la definición

`(defun f (x1 ... xn) cuerpo)`

aseguran que existe una *medida de terminación* m en el lenguaje de $\mathcal{H}$, respecto de la cual se tienen

- m es la representación en ACL2 de un ordinal

`(o-p m)`

- Por cada ocurrencia en `cuerpo` de un subtérmino `(f u1 ... un)` se puede demostrar la fórmula

`(implies (and t1 ... tk)
 (o< $\sigma(m)$ m))`

donde $t_1, \dots, t_k$ son los términos que influyen en dicha ocurrencia y σ es la sustitución $\{x_1/u_1, \dots, x_n/un\}$

- La función factorial

```
(defun factorial (n)
  (if (zp n)
      1
      (* n (factorial (- n 1)))))
```

- Medida de terminación $m = (\text{if } (\text{natp } n) \ n \ 0)$
 - Hay una única llamada recursiva $(\text{factorial } (- \ n \ 1))$ y el término que influye en ella es $(\text{not } (\text{zp } n))$
 - La conjetura de terminación es

```
(implies (not (zp n))
  (o< (if (natp (- n 1)) (- n 1) 0)
    (if (natp n) n 0)))
```

- Concatenación de listas

```
(defun concatena (l1 l2)
  (if (endp l1)
      l2
      (cons (car l1) (concatena (cdr l1) l2))))
```

- Medida de terminación $m = (\text{len } l1)$

- Hay una única llamada recursiva `(concatena (cdr l1) l2)` y el término que influye en ella es `(not (endp l1))`
- La conjetura de terminación es

```
(implies (not (endp l1))
  (o< (len (cdr l1)) (len l1)))
```

- Listas ordenadas

```
(defun ordenada (l)
  (cond ((endp l) (equal l nil))
        (t (or (endp (cdr l))
                 (and (<= (car l) (cadr l))
                      (ordenada (cdr l)))))))
```

- Medida de terminación $m = (\text{len } l)$

- Hay una única llamada recursiva `(ordenada (cdr l))` y los términos que influyen en ella son `(not (endp l))`, `(not (endp (cdr l)))` y `(<= (car l) (cadr l))`
- La conjetura de terminación es

```
(implies (and (not (endp l))
               (not (endp (cdr l)))
               (<= (car l) (cadr l)))
          (o< (len (cdr l)) (len l)))
```

- Mezcla ordenada de dos listas

```
(defun mezcla (l1 l2)
  (cond ((endp l1) l2)
        ((endp l2) l1)
        ((< (car l1) (car l2))
         (cons (car l1) (mezcla (cdr l1) l2)))
        (t (cons (car l2) (mezcla l1 (cdr l2))))))
```

- Medida de terminación $m = (+ (\text{len } l1) (\text{len } l2))$
- Hay dos llamadas recursivas
 - `(mezcla (cdr l1) l2)`
 - `(mezcla l1 (cdr l2))`

- La primera llamada recursiva es `(mezcla (cdr 11) 12)` y los términos que influyen en ella son `(not (endp 11))`, `(not (endp 12))` y `(< (car 11) (car 12))`
- La conjetura de terminación es

```
(implies (and (not (endp 11))
              (not (endp 12))
              (< (car 11) (car 12)))
  (o< (+ (len (cdr 11)) (len 12))
      (+ (len 11) (len 12))))
```

- La segunda llamada recursiva es `(mezcla 11 (cdr 12))` y los términos que influyen en ella son `(not (endp 11))`, `(not (endp 12))` y `(not (< (car 11) (car 12)))`
- La conjetura de terminación es

```
(implies (and (not (endp 11))
              (not (endp 12))
              (not (< (car 11) (car 12))))
  (o< (+ (len 11) (len (cdr 12)))
      (+ (len 11) (len 12))))
```

- Usualmente el sistema es capaz de deducir una medida de terminación adecuada a partir de los casos recursivos de la definición
- Para proporcionar explícitamente la medida de terminación utilizamos la directiva `(declare (xargs :measure m))`

```
(defun mezcla (l1 l2)
  (declare (xargs :measure (+ (len l1) (len l2))))
  (cond ((endp l1) l2)
        ((endp l2) l1)
        ((< (car l1) (car l2))
         (cons (car l1) (mezcla (cdr l1) l2)))
        (t (cons (car l2) (mezcla l1 (cdr l2))))))
```

El principio de inducción estructural en ACL2

- Una prueba por inducción de $\phi \equiv \sum_{k=1}^n (2k - 1) = n^2$
- En ACL2: $\phi \equiv (\text{equal } (\text{sum-2k-1 } n) (* n n))$
 - Caso base: $n \notin \mathbb{N}^+ \rightarrow \phi(n)$

```
(implies (zp n)
         (equal (sum-2k-1 n) (* n n)))
```

- Caso de inducción: $((n \in \mathbb{N}^+) \wedge \phi(n - 1)) \rightarrow \phi(n)$

```
(implies (and (not (zp n))
              (equal (sum-2k-1 (- n 1))
                    (* (- n 1) (- n 1))))
         (equal (sum-2k-1 n) (* n n)))
```

- La inducción es válida pues la hipótesis de inducción se establece para el valor $(- n 1)$, menor que n , y el conjunto de los números naturales está bien ordenado con respecto al orden usual

El principio de inducción estructural en ACL2

- Una fórmula ϕ se deduce a partir de las siguientes fórmulas
 - *Casos de inducción*: Para cada $1 \leq i \leq k$

```
(implies (and  $q_i$   $\sigma_{i,1}(\phi)$  ...  $\sigma_{i,h_i}(\phi)$ )  $\phi$ )
```

- *Caso base*

```
(implies (and (not  $q_1$ ) ... (not  $q_k$ ))  $\phi$ )
```

donde $q_1, \dots, q_k$ son términos, $\sigma_{i,j}$, ($1 \leq i \leq k, 1 \leq j \leq h_i$) son sustituciones y, para cierto término m , se tienen

- m es la representación en ACL2 de un ordinal

```
(o-p  $m$ )
```

- Para cada i, j tales que $1 \leq i \leq k$ y $1 \leq j \leq h_i$

```
(implies  $q_i$  (o<  $\sigma_{i,j}(m)$   $m$ ))
```

- Decimos que ϕ se demuestra por *inducción en las variables* del término m , denominado *medida*

El principio de inducción estructural en ACL2

- Una prueba por inducción de $\phi \equiv \sum_{k=1}^n (2k - 1) = n^2$
- Caso de inducción: $q_1 \equiv (\text{not } (zp\ n)), \sigma_{1,1} = \{n / (-\ n\ 1)\}$

```
(implies (and (not (zp n))
              (equal (sum-2k-1 (- n 1))
                    (* (- n 1) (- n 1))))
         (equal (sum-2k-1 n) (* n n)))
```

- Caso base

```
(implies (zp n)
         (equal (sum-2k-1 n) (* n n)))
```

- Medida de inducción: $m = (\text{if } (\text{natp } n) \ n \ 0)$

```
(o-p (if (natp n) n 0))

(implies (not (zp n))
         (o< (if (natp (- n 1)) (- n 1) 0)
            (if (natp n) n 0)))
```

Inducción sugerida por una definición

- Una prueba por inducción de $\phi \equiv \sum_{k=1}^n (2k - 1) = n^2$
- Definición en ACL2

```
(defun sum-2k-1 (n)
  (cond ((zp n) 0)
        (t (+ (- (* 2 n) 1)
               (sum-2k-1 (- n 1))))))
```

- Medida de terminación $m = (\text{if } (\text{natp } n) \ n \ 0)$
- Hay una única llamada recursiva $(\text{sum-2k-1 } (- \ n \ 1))$ y el término que influye en ella es $(\text{not } (\text{zp } n))$
- La conjetura de terminación es

```
(implies (not (zp n))
  (o< (if (natp (- n 1)) (- n 1) 0)
      (if (natp n) n 0)))
```

Inducción sugerida por una definición

- Una prueba por inducción de $\phi \equiv \sum_{k=1}^n (2k - 1) = n^2$
- Caso de inducción: $t_{1,1} \equiv (\text{not } (\text{zp } n)), \sigma_1 = \{n / (-\ n\ 1)\}$

```
(implies (and (not (zp n))  
              (equal (sum-2k-1 (- n 1))  
                    (* (- n 1) (- n 1))))  
         (equal (sum-2k-1 n) (* n n)))
```

- Caso base:

```
(implies (zp n)  
         (equal (sum-2k-1 n) (* n n)))
```

- Medida de inducción $m = (\text{if } (\text{natp } n) \ n\ 0)$
- Las propiedades de la medida de inducción coinciden con las conjeturas de terminación

Inducción sugerida por una definición

- Consideremos la definición `(defun f (x1 ... xn) cuerpo)`, con m casos finales en `cuerpo` en los que intervienen n_m llamadas recursivas de la forma `(f u1i ... uni)` y siendo $t_{i,1}, \dots, t_{i,k_i}$ los términos que influyen en dicho caso final
- Una fórmula ϕ se deriva a partir de las siguientes fórmulas
 - Caso base:*

```
(implies (and (not (and t1,1 ... t1,k1))  
              ...  
              (not (and tm,1 ... tm,km)))  $\phi$ )
```

- Casos de inducción:* Por cada llamada recursiva en `cuerpo`

```
(implies (and ti,1 ... ti,ki  
               $\sigma_{i,1}(\phi)$  ...  $\sigma_{i,n_i}(\phi)$ )  $\phi$ )
```

donde $\sigma_{i,j}$ es la sustitución $\{x_1/u_{1i}, \dots, x_n/u_{ni}\}$ correspondiente a la j -ésima llamada recursiva en el i -ésimo caso

Inducción sugerida por una definición

- Las conjeturas de terminación de la definición

`(defun f (x1 ... xn) cuerpo)`

aseguran que existe una *medida de terminación* m que cumple

- m es la representación en ACL2 de un ordinal:

`(o-p m)`

- por cada llamada recursiva en `cuerpo`

`(implies (and ti,1 ... ti,ki)
 (o< $\sigma_{i,j}(m)$ m))`

- No es necesario que la función f aparezca en la conjetura a demostrar

Ejemplo de inducción sugerida por una definición

- Conjetura

```
(equal (len (concatena 11 12))  
      (+ (len 11) (len 12)))
```

- Posibles inducciones sugeridas por las funciones implicadas en la conjetura
 - Inducción sugerida por (concatena 11 12)
 - Inducción sugerida por (len 11)
 - Inducción sugerida por (len 12)

Ejemplo de inducción sugerida por una definición

- Definiciones

```
(defun len (l)
  (if (consp l)
      (+ 1 (len (cdr l)))
      0))

(defun concatena (l1 l2)
  (if (endp l1)
      l2
      (cons (car l1) (concatena (cdr l1) l2))))
```


Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(concatena 11 12)`
 - Caso base: $t_{1,1} \equiv (\text{not } (\text{endp } 11))$

```
(implies (endp 11)
  (equal (len (concatena 11 12))
    (+ (len 11) (len 12))))
```

- Caso de inducción: $\sigma_1 = \{11 / (\text{cdr } 11)\}$

```
(implies (and (not (endp 11))
  (equal (len (concatena (cdr 11) 12))
    (+ (len (cdr 11)) (len 12))))
  (equal (len (concatena 11 12))
    (+ (len 11) (len 12))))
```

- Medida de inducción: $m = (\text{len } 11)$

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(len 11)`

- Caso base: $t_{1,1} \equiv (\text{consp } 11)$

```
(implies (not (consp 11))  
         (equal (len (concatena 11 12))  
                (+ (len 11) (len 12))))
```

- Caso de inducción: $\sigma_1 = \{11 / (\text{cdr } 11)\}$

```
(implies (and (consp 11)  
              (equal (len (concatena (cdr 11) 12))  
                     (+ (len (cdr 11)) (len 12))))  
         (equal (len (concatena 11 12))  
                (+ (len 11) (len 12))))
```

- Medida de inducción: $m = (\text{len } 11)$
- Equivalente a la inducción sugerida por `(concatena 11 12)`

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por $(\text{len } 12)$

- Caso base: $t_{1,1} \equiv (\text{cons } p \ 12)$

```
(implies (not (cons p 12))  
         (equal (len (concatena 11 12))  
                (+ (len 11) (len 12))))
```

- Caso de inducción: $\sigma_1 = \{12 / (\text{cdr } 12)\}$

```
(implies (and (cons p 12)  
              (equal (len (concatena 11 (cdr 12)))  
                     (+ (len 11) (len (cdr 12)))))  
         (equal (len (concatena 11 12))  
                (+ (len 11) (len 12))))
```

- Medida de inducción: $m = (\text{len } 12)$

Ejemplo de inducción sugerida por una definición

- Conjetura

```
(implies (and (ordenada 11)
              (ordenada 12))
         (ordenada (mezcla 11 12)))
```

- Posibles inducciones sugeridas por las funciones implicadas en la conjetura
 - Inducción sugerida por `(ordenada 11)`
 - Inducción sugerida por `(ordenada 12)`
 - Inducción sugerida por `(mezcla 11 12)`

Ejemplo de inducción sugerida por una definición

- Definiciones

```
(defun ordenada (l)
  (cond ((endp l) (equal l nil))
        (t (or (endp (cdr l))
                (and (<= (car l) (cadr l))
                     (ordenada (cdr l)))))))

(defun mezcla (l1 l2)
  (cond ((endp l1) l2)
        ((endp l2) l1)
        ((< (car l1) (car l2))
         (cons (car l1) (mezcla (cdr l1) l2)))
        (t (cons (car l2) (mezcla l1 (cdr l2))))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(ordenada l1)`
 - Medida de inducción: $m = (\text{len } l1)$
 - Caso de inducción: $\sigma_1 = \{l1 / (\text{cdr } l1)\}$

```
(implies
  (and (not (endp l1))                                ; t1,1
        (not (endp (cdr l1)))                          ; t1,2
        (<= (car l1) (cadr l1))                        ; t1,3
        (implies (and (ordenada (cdr l1))
                      (ordenada l2))
                  (ordenada (mezcla (cdr l1) l2))))
  (implies (and (ordenada l1)
                (ordenada l2))
            (ordenada (mezcla l1 l2))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (ordenada 11)
 - Caso base

```
(implies
  (not (and (not (endp 11))
            (not (endp (cdr 11)))
            (<= (car 11) (cadr 11))))      ; t1,1
  (implies (and (ordenada 11)
                (ordenada 12))
            (ordenada (mezcla 11 12))))    ; t1,2
                                           ; t1,3
```

- Comportamiento del or computacional

$$\begin{aligned} & (\text{not } (\text{and } q_1 \dots q_m)) \\ & \equiv (\text{or } (\text{not } q_1) \dots (\text{not } q_m)) \\ & \equiv \begin{cases} (\text{not } q_1) \\ (\text{and } q_1 (\text{not } q_2)) \\ \dots \\ (\text{and } q_1 \dots q_{m-1} (\text{not } q_m)) \end{cases} \end{aligned}$$

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (ordenada 11)
 - Casos bases

```
(implies
  (endp 11)                                     ;  $\neg t_{1,1}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))

(implies
  (and (not (endp 11))                         ;  $t_{1,1}$ 
        (endp (cdr 11)))                     ;  $\neg t_{1,2}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))

(implies
  (and (not (endp 11))                         ;  $t_{1,1}$ 
        (not (endp (cdr 11)))                 ;  $t_{1,2}$ 
        (not (<= (car 11) (cadr 11))))       ;  $\neg t_{1,3}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))
```


Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(ordenada 12)`
 - Medida de inducción: $m = (\text{len } 12)$
 - Caso de inducción: $\sigma_1 = \{12 / (\text{cdr } 12)\}$

```
(implies
  (and (not (endp 12))                                ; t1,1
        (not (endp (cdr 12)))                          ; t1,2
        (<= (car 12) (cadr 12))                        ; t1,3
        (implies (and (ordenada 11)
                       (ordenada (cdr 12)))
                  (ordenada (mezcla 11 (cdr 12)))))
  (implies (and (ordenada 11)
                (ordenada 12))
            (ordenada (mezcla 11 12))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (ordenada 12)
 - Casos bases

```
(implies
  (endp 12)                                     ;  $\neg t_{1,1}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))

(implies
  (and (not (endp 12))                          ;  $t_{1,1}$ 
        (endp (cdr 12)))                       ;  $\neg t_{1,2}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))

(implies
  (and (not (endp 12))                          ;  $t_{1,1}$ 
        (not (endp (cdr 12)))                  ;  $t_{1,2}$ 
        (not (<= (car 12) (cadr 12))))         ;  $\neg t_{1,3}$ 
  (implies (and (ordenada 11)
                 (ordenada 12))
            (ordenada (mezcla 11 12))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(mezcla 11 12)`
 - Medida de inducción: $m = (+ \text{ (len 11) (len 12)})$
 - Caso de inducción: $\sigma_1 = \{11 / (\text{cdr } 11)\}$

```
(implies
  (and (not (endp 11)) ; t1,1
        (not (endp 12)) ; t1,2
        (< (car 11) (car 12)) ; t1,3
        (implies (and (ordenada (cdr 11))
                       (ordenada 12))
                  (ordenada (mezcla (cdr 11) 12))))
  (implies (and (ordenada 11)
                (ordenada 12))
            (ordenada (mezcla 11 12))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(mezcla 11 12)`
 - Caso de inducción: $\sigma_2 = \{12 / (\text{cdr } 12)\}$

```
(implies
  (and (not (endp 11)) ; t2,1
        (not (endp 12)) ; t2,2
        (not (< (car 11) (car 12))) ; t2,3
        (implies (and (ordenada 11)
                       (ordenada (cdr 12)))
                  (ordenada (mezcla 11 (cdr 12)))))
  (implies (and (ordenada 11)
                (ordenada 12))
            (ordenada (mezcla 11 12))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (mezcla 11 12)
 - Caso base

```
(implies
  (and (not (and (not (endp 11))                ; t1,1
                 (not (endp 12))                ; t1,2
                 (< (car 11) (car 12))))         ; t1,3
    (not (and (not (endp 11))                    ; t2,1
              (not (endp 12))                    ; t2,2
              (<= (car 12) (car 11)))))          ; t2,3
  (implies (and (ordenada 11)
                (ordenada 12))
    (ordenada (mezcla 11 12))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (mezcla 11 12)
 - Caso base

```
(implies (endp 11)                                ;  $\neg t_{1,1}$ 
  (implies (and (ordenada 11)
                (ordenada 12))
    (ordenada (mezcla 11 12))))

(implies (and (not (endp 11))                      ;  $t_{1,1}$ 
              (endp 12))                          ;  $\neg t_{2,2}$ 
  (implies (and (ordenada 11)
                (ordenada 12))
    (ordenada (mezcla 11 12))))
```

- Correspondencia con los casos de la definición de mezcla

Ejemplo de inducción sugerida por una definición

- Definiciones

```
(defun len (l)
  (if (consp l)
      (+ 1 (len (cdr l)))
      0))

(defun hojas (tree)
  (if (consp tree)
      (if (consp (car tree))
          (+ (hojas (car tree)) (hojas (cdr tree)))
          (+ 1 (hojas (cdr tree))))
      0))

(defun flat (tree)
  (if (consp tree)
      (if (consp (car tree))
          (append (flat (car tree)) (flat (cdr tree)))
          (cons (car tree) (flat (cdr tree))))
      nil))
```

Ejemplo de inducción sugerida por una definición

- Conjetura

```
(equal (hojas tree)  
      (len (flat tree)))
```

- Posibles inducciones sugeridas por las funciones implicadas en la conjetura
 - Inducción sugerida por `(hojas tree)`
 - Inducción sugerida por `(flat tree)`
- Ambos esquemas de inducción son iguales

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (hojas tree)
 - Primer caso de inducción: $\sigma_{1,1} = \{\text{tree}/(\text{car tree})\}$
 $\sigma_{1,2} = \{\text{tree}/(\text{cdr tree})\}$

```
(implies
  (and (consp tree)
        (consp (car tree))
        (equal (hojas (car tree))
                 (len (flat (car tree))))           ;  $\sigma_{1,1}(\phi)$ 
        (equal (hojas (cdr tree))
                 (len (flat (cdr tree))))))         ;  $\sigma_{1,2}(\phi)$ 
  (equal (hojas tree)
          (len (flat tree))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por (hojas tree)
 - Segundo caso de inducción: $\sigma_{2,1} = \{\text{tree}/(\text{cdr tree})\}$

```
(implies
  (and (consp tree)
        (not (consp (car tree)))
        (equal (hojas (cdr tree))
                (len (flat (cdr tree))))) ;  $\sigma_{2,1}(\phi)$ 
  (equal (hojas tree)
          (len (flat tree))))
```

Ejemplo de inducción sugerida por una definición

- Inducción sugerida por `(hojas tree)`
- Caso base:

```
(implies (not (consp tree))  
         (equal (hojas tree)  
                (len (flat tree))))
```

- Kaufmann, M. and Manolios, P. and Moore, J S.
Computer-Aided Reasoning: An Approach. (Kluwer Academic Publishers, 2000)
- Kaufmann, M. and Moore, J S. *ACL2 Version 8.5*.
(<http://www.cs.utexas.edu/users/moore/acl2/>)